

# CROSSTAB Example #9

## *SUDAAN Statements and Results Illustrated*

- Kappa measure of agreement
- AGREE statement
- TABLE statement
- MERGHI option
- WSUM option

## *Input Data Set(s): NHANES3S3.SAS7bdat*

### *Example*

*Using NHANES III periodontal data, estimate the Kappa measure of intra-dental examiner agreement between ratings of baseline vs. follow-up periodontal assessments of teeth.*

### *Solution*

The dataset for this example consists of the approximately 5% of the examined survey sample of *NHANES III* recalled for repeat second dental exams identical to the first. These sample persons were examined twice by the same examining dentist at an interval of 1—6 weeks after the initial exam (Winn, et al 1999). The data from repeat exams on the sub-sample are available as a public use electronic dataset. Intra-examiner reliability is assessed using the Kappa measure of agreement on the repeat visits (baseline and follow-up) for each sample member. In this example, the periodontal outcome of interest is *Probing Pocket Depth Score, upper central incisor, mesial surface* (measured in millimeters).

Recall that *NHANES III* is a cross-sectional sample survey of the civilian, non-institutionalized population aged 2 months or older, fielded during 1988-1994. The sample weight variable is WTPFQX6, and the stratification and PSU variables are SDPSTRA6 and SDPPSU6, respectively.

The SAS-Callable SUDAAN code for this example is displayed in *Exhibit 1*. The square table to be used for estimation of Kappa is defined by the last two variables on the AGREE and TABLES statement—gender (*hssex*) by baseline (*depumpc1s*) by follow-up (*derumpc1s*). The AGREE statement requests SUDAAN to estimate kappa for each level of gender (*hssex*) and overall. The TABLES statement is not needed to estimate Kappa but is added here in order to display the sample size and estimated population size in each cell for descriptive purposes.

When requesting the Kappa measure of agreement, the RxC variables that define rows and columns in the table must be specified on the SUBGROUP and LEVELS statements. Stratification variables can appear on either the SUBGROUP or CLASS statements, and in this example the stratification variable is included on the CLASS statement.

The PRINT statement instructs SUDAAN to print sample size (NSUM) and estimated population size (WSUM) in each cell of the stratified table, and then to print point estimates and 95% confidence limits of Kappa and a test for  $H_0: \text{Kappa}=0$  for each of the stratification levels specified on the AGREE statement.

This example was run in SAS-Callable SUDAAN, and the SAS program and \*.LST files are provided.

## Exhibit 1. SAS-Callable SUDAAN Code

```
* This program calculates the KAPPA statistic using the NHANES III periodontal data;

LIBNAME dent "\\rtints29\sudaan\data\nhanes3\dental";

PROC FORMAT;
    value gen_ 1="Male"
             2="Female";
    value mm 1="0 mm"
            2="1 mm"
            3="2 mm"
            4="3+ mm";

PROC CROSSTAB DATA=dent.dental2 NOTSORTED MERGEHI;
    NEST sdpstra6 sdpps6;
    WEIGHT wtpfqx6;

    SUBGROUP depumpcls derumpcls;
    LEVEL    4          4;

    CLASS HSSEX;
    TABLES hssex*depumpcls*derumpcls;
    AGREE   hssex*depumpcls*derumpcls;

    SETENV ROWWIDTH=10 COLWIDTH=6 COLSPCE=1 TOPMGN=0 lblwidth=8;
    PRINT NSUM="SamSize" / nsumfmt=f10.0;

    SETENV ROWWIDTH=10 COLWIDTH=6 COLSPCE=1 TOPMGN=0 lblwidth=8;
    PRINT WSUM="PopSize" / wsumfmt=f10.0;

    SETENV COLWIDTH=10 COLSPCE=1 TOPMGN=0 LABWIDTH=12;
    PRINT / KAPPA=default KTEST=default lowkfmt=f9.4 upkfmt=f9.4 ;

    RFORMAT hssex gen_.;
    rformat depumpcls mm.;
    rformat derumpcls mm.;

    RLABEL depumpcls="Baseline: Probing Pocket Depth Score, upper central incisor,
                    mesial (mm)";
    RLABEL derumpcls="Follow up: Probing Pocket Depth Score, upper central incisor,
                    mesial (mm)";

    RTITLE "Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments
           of Teeth";
    RFOOTNOTE "Data Source: NHANES III, 1988-1994";
```

Note the option MERGEHI on the PROC statement. The outcomes in the raw data range from 0—12mm, with the vast majority between 0—3mm. The MERGEHI option includes all values above 3mm in the same category as 3mm. Therefore, the last category of probing pocket depth score is 3mm or greater. Kappa can only be computed for square tables.

The probing pocket depth outcome that is used in analysis effectively has 4 levels, defined as follows (values that are coded “could not be assessed” are set to missing):

- 1=0 mm
- 2=1 mm
- 3=2 mm
- 4=3+ mm

## Exhibit 2. First Page of SUDAAN Output (SAS \*.LST File)

```

                S U D A A N
      Software for the Statistical Analysis of Correlated Data
      Copyright      Research Triangle Institute      May 2011
                Release 11.0.0

DESIGN SUMMARY: Variances will be computed using the Taylor Linearization Method, Assuming a
With Replacement (WR) Design
      Sample Weight: WTPFQX6
      Stratification Variables(s): SDPSTRA6
      Primary Sampling Unit: SDPPSU6

Number of observations read      :    2596      Weighted count : 24789861
Denominator degrees of freedom :      49

```

There are 2,596 records (sample persons) on the file, representing 24,789,861 people in the population.

## Exhibit 3. CLASS Variable Frequencies

```

Frequencies and Values for CLASS Variables
by: Sex.

-----
Sex          Frequency      Value
-----
Ordered
  Position:
    1                1204      Male
Ordered
  Position:
    2                1392      Female
-----

```

Males and females are roughly evenly divided in the dataset.

**Exhibit 4. Baseline\*Follow-up Crosstabulation (Sex=Total)**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Total.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)		Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	SamSize	969	32	514	310	113
0 mm	SamSize	21	<b>13</b>	7	1	0
1 mm	SamSize	528	17	<b>415</b>	88	8
2 mm	SamSize	301	2	85	<b>185</b>	29
3+ mm	SamSize	119	0	7	36	<b>76</b>

Data Source: NHANES III, 1988-1994

A total of 969 observations have both a baseline and follow-up dental exam that could be assessed (the balance of the observations on the file were persons that had at least one time point in which the outcome could not be assessed). Observations in cells along the diagonal represent instances of agreement between baseline and follow-up exam—there was 71% agreement between baseline and follow-up using unweighted data.

**Exhibit 4. Baseline\*Follow-up Crosstabulation (Sex=Total) cont'd.**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Total.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)	PopSize	Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	9966690	222377	5659305	3276395	808613	
0 mm	216739	<b>87469</b>	106595	22675	0	
1 mm	6042946	117001	<b>4805563</b>	1057887	62496	
2 mm	2887638	17907	697591	<b>1906901</b>	265239	
3+ mm	819367	0	49556	288933	<b>480878</b>	

Data Source: NHANES III, 1988-1994

This table represents the weighted totals in each cell. The 969 observations with complete data for baseline and follow-up represent 9,966,690 people in the population. Observations in cells along the diagonal represent instances of agreement between baseline and follow-up exam—there was 73% agreement between baseline and follow-up exams using weighted data. Kappa is computed from the weighted survey totals in this table.

**Exhibit 5. Baseline\*Follow-up Crosstabulation (Sex=Male)**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Male.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)	SamSize	Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	SamSize	450	6	218	162	64
0 mm	SamSize	4	<b>2</b>	1	1	0
1 mm	SamSize	226	4	<b>178</b>	39	5
2 mm	SamSize	153	0	37	<b>101</b>	15
3+ mm	SamSize	67	0	2	21	<b>44</b>

Data Source: NHANES III, 1988-1994

This table contains the sample size in each cell for Males.

**Exhibit 5. Baseline\*Follow-up Crosstabulation (Sex=Male) cont'd.**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Male.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)	PopSize	Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	4654498	50374	2417736	1742454	443935	
0 mm	40795	<b>15501</b>	2619	22675	0	
1 mm	2659728	34873	<b>2051096</b>	530940	42819	
2 mm	1495421	0	341262	<b>1020223</b>	133936	
3+ mm	458555	0	22759	168615	<b>267180</b>	

Data Source: NHANES III, 1988-1994

This table contains the weighted totals in each cell for Males.

**Exhibit 6. Baseline\*Follow-up Crosstabulation (Sex=Female)**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Female.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)	SamSize	Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	519	26	296	148	49	
0 mm	17	<b>11</b>	6	0	0	
1 mm	302	13	<b>237</b>	49	3	
2 mm	148	2	48	<b>84</b>	14	
3+ mm	52	0	5	15	<b>32</b>	

Data Source: NHANES III, 1988-1994

This table contains the sample size in each cell for Females.



**Exhibit 6. Baseline\*Follow-up Crosstabulation (Sex=Female) cont'd.**

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

by: Sex, Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm), Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm).

for: Sex = Female.

Baseline: Probing Pocket Depth Score, upper central incisor, mesial (mm)	PopSize	Follow up: Probing Pocket Depth Score, upper central incisor, mesial (mm)				
		Total	0 mm	1 mm	2 mm	3+ mm
Total	5312191	172004	3241568	1533942	364678	
0 mm	175944	<b>71968</b>	103976	0	0	
1 mm	3383219	82129	<b>2754467</b>	526947	19677	
2 mm	1392217	17907	356329	<b>886677</b>	131303	
3+ mm	360812	0	26797	120318	<b>213698</b>	

Data Source: NHANES III, 1988-1994

This table contains the weighted totals in each cell for Females.

## Exhibit 7. Stratum-Specific Kappa Estimates

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

Kappa Agreement Statistics  
Variable DEPUMPC1S by Variable DERUMPC1S

by: Sex.

Sex	Kappa	SE	Lower 95% Limit	Upper 95% Limit
Total	0.4864	0.0336	0.4189	0.5539
Male	0.4772	0.0620	0.3527	0.6017
Female	0.4896	0.0516	0.3860	0.5932

Data Source: NHANES III, 1988-1994

This table contains the survey-weighted Kappa estimates with 95% confidence limits, within stratification level (gender) and overall. Agreement is close to 0.48 in all cases, indicating moderate agreement between repeat exams. And we can also conclude that reliability between baseline and follow-up exams did not differ across population subgroups defined by gender.

### Exhibit 8. Stratum-Specific Test of H0: Kappa=0

Variance Estimation Method: Taylor Series (WR)

Intra-Examiner Agreement Baseline vs. Follow-up Periodontal Assessments of Teeth

Test Statistics for the Agreement Null Hypothesis H0: Kappa=0  
Variable DEPUMPC1S by Variable DERUMPC1S

by: Sex.

Sex	DDF	T-Test	P-Value
Total	49	14.4707	0.0000
Male	49	7.7017	0.0000
Female	49	9.4963	0.0000

Data Source: NHANES III, 1988-1994

This table contains the *t*-test for H<sub>0</sub>: Kappa=0 for the overall population and for each gender. The confidence limits on the previous page did not contain a value of 0, and hence the *t*-test is significant (*p*=0.0000) in each case.

Note that the same results could have been obtained by specifying HSSEX on a BY statement instead of a stratification variable on the TABLES and AGREE statements:

### Exhibit 9. SAS-Callable SUDAAN Code: BY HSSEX

```
PROC CROSSTAB DATA=dent.dental2 NOTSORTED MERGEHI;
  NEST sdpstra6 sdpps6;
  WEIGHT wtpfqx6;

  SUBGROUP depumpcls derumpcls;
  LEVEL 4 4;

  RBY HSSEX;
  TABLES depumpcls*derumpcls;
  AGREE depumpcls*derumpcls;

  SETENV ROWWIDTH=10 COLWIDTH=6 COLSPACE=1 TOPMGN=0 lblwidth=8;
  PRINT NSUM="SamSize" / nsumfmt=f10.0;

  SETENV ROWWIDTH=10 COLWIDTH=6 COLSPACE=1 TOPMGN=0 lblwidth=8;
  PRINT WSUM="PopSize" / wsumfmt=f10.0;

  SETENV COLWIDTH=10 COLSPACE=1 TOPMGN=0 LABWIDTH=12;
  PRINT / KAPPA=default KTEST=default lowkfmt=f9.4 upkfmt=f9.4 ;
  RTITLE "Kappa with BY Statement";
```